

Zadanie 25.

Litera A

Matura 2006. Poziom rozszerzony

Wykorzystane funkcje: CZY.LICZBA, JEŻELI, LEWY, LICZ.JEŻELI, PRAWY, SZUKAJ.TEKST

Wykorzystane narzędzia: funkcje, import danych

Poziom trudności: 4/5

Treść zadania

W pliku *hasla.txt* umieszczono 200 wyrazów zapisanych małymi literami. Policz, ile wyrazów rozpoczyna się na literę „a”, ile na nią kończy, a ile zawiera ją w środku (nie na pierwszej bądź ostatniej pozycji).

Rozwiązanie

Import danych

1. W pliku *hasla.txt* w folderze 25 w każdej linii znajduje się jeden wyraz. Ponieważ plik nie zawiera nagłówka kolumny, do komórki *A1* wpisz wyraz. Kolejne kolumny będą zawierały:
 - a) Sprawdzenie, czy pierwsza litera z wyrazu umieszczonego w kolumnie *A* to litera „a” — do komórki *B1* wpisz więc pierwsza.
 - b) Sprawdzenie, czy ostatnia litera to „a” — do komórki *C1* wpisz więc ostatnia.
 - c) Sprawdzenie, czy środkowa część wyrazu zawiera literę „a” — do komórki *D1* wpisz więc środek.

2. Aby zaimportować hasła z pliku *hasla.txt*, na karcie *Dane* wybierz polecenie *Z tekstu* i wskaż plik *hasla.txt*.
 - a) W każdej linii umieszczone jest tylko jedno hasło, więc w pierwszym kroku kreatora importu tekstu zaznacz *Rozdzielany* i kliknij *Dalej*.
 - b) W kroku drugim pozostaw zaznaczoną opcję *Tabulator* i kliknij *Dalej*.
 - c) W ostatnim kroku kreatora upewnij się, że *Typ danych w kolumnie* został ustawiony na *Ogólny*, i kliknij *Zakończ*.
 - d) Zaznacz komórkę *A2* jako komórkę, od której ma się zacząć wyświetlanie kolejnych liczb, i kliknij *OK*.

	A	B	C	D
1	wyraz	pierwsza	ostatnia	środek
2	adghgd			
3	ajakok			
4	akokos			
5	amkoku			
6	amodda			
7	aokor			
8	aokormn			
9	apraw			
10	asprz			

Funkcje tekstowe

1. W celu sprawdzenia, czy wyraz zaczyna się od litery „a”, wstaw do komórki *B2* funkcję *JEŻELI*. Jako *test logiczny* należy wpisać wyrażenie, dzięki któremu najpierw wytniesz pierwszą literę (z lewej strony), a następnie sprawdzisz, czy jest to litera „a”. Wstaw funkcję *LEWY*, jako *tekst* zaznacz komórkę *A2*, a jako *liczba* wpisz 1. Dopisz w teście logicznym `= "a"` w celu sprawdzenia, czy litera jest równa „a”. Cały *test logiczny* to `LEWY(A2;1)="a"`. Jeżeli test logiczny jest spełniony, wówczas powinna być wyświetlana wartość *PRAWDA*; jeżeli nie jest spełniony — wartość *FAŁSZ*. Cała składnia funkcji w komórce *B2* to `=JEŻELI(LEWY(A2;1)="a";PRAWDA;FAŁSZ)`.
2. W analogiczny sposób należy sprawdzić, czy wyraz kończy się na literę „a”. Do komórki *C2* ponownie wstaw funkcję *JEŻELI*. Tym razem w teście logicznym użyj funkcji *PRAWY*, gdyż należy sprawdzić ostatnią literę wyrazu, czyli pierwszą z prawej strony. Funkcja w komórce *C2* to `=JEŻELI(PRAWY(A2;1)="a";PRAWDA;FAŁSZ)`.
3. Aby sprawdzić, czy w wyrazie w środku znajduje się litera „a”, wstaw do komórki *D2* funkcję *SZUKAJ.TEKST*. Jako *szukany_tekst* wpisz `"?a*?"`, gdzie znaki zapytania oznaczają, że nie będzie sprawdzana pierwsza i ostatnia litera, a gwiazdka oznacza dowolny ciąg znaków. Jako parametr *obejmujący_tekst* wpisz *A2*, gdyż to w wyrazie znajdującym się w tej komórce ma być szukany ciąg znaków. W wyniku działania tej funkcji otrzymasz wartość liczbową, gdy tekst zostanie znaleziony w komórce *A2*, bądź wartość *ARG#*, gdy tekstu

nie ma. Jeżeli w kolumnie *D* chcesz otrzymać wartość PRAWDA lub FAŁSZ, obejmij całą funkcję SZUKAJ.TEKST funkcją CZY.LICZBA. Cały wpis w komórce *D2* to =CZY.LICZBA(SZUKAJ.TEKST("?*a*?";A2)).

- Przeciagnij formuły z komórek *B2*, *C2* i *D2* dla pozostałych wyrazów, czyli do wiersza *201*.

	A	B	C	D
1	wyraz	pierwsza	ostatnia	środek
2	adghgd	PRAWDA	FAŁSZ	FAŁSZ
3	ajakok	PRAWDA	FAŁSZ	PRAWDA
4	akokos	PRAWDA	FAŁSZ	FAŁSZ
5	amkoku	PRAWDA	FAŁSZ	FAŁSZ
6	amodda	PRAWDA	PRAWDA	FAŁSZ
7	aokor	PRAWDA	FAŁSZ	FAŁSZ
8	aokormn	PRAWDA	FAŁSZ	FAŁSZ
9	apraw	PRAWDA	FAŁSZ	PRAWDA
10	asprz	PRAWDA	FAŁSZ	FAŁSZ

Funkcje matematyczne

Ostatnim krokiem jest policzenie, ile razy w każdej z kolumn (*B*, *C* i *D*) pojawiła się wartość PRAWDA. Wpisz do komórek *F2*, *F3* i *F4* kolejno: "a" pierwsze, "a" końcowe i "a" środkowe.

- Litera „a” jako pierwsza. Do komórki *G2* wstaw funkcję LICZ.JEŻELI. Jako parametr *zakres* zaznacz całą kolumnę *B*, a jako kryteria wpisz "PRAWDA". Cały wpis w komórce to =LICZ.JEŻELI(B:B;PRAWDA).
- Litera „a” jako ostatnia. Do komórki *G3* wstaw ponownie funkcję LICZ.JEŻELI. Tym razem jako parametr *zakres* zaznacz kolumnę *C*, a kryteria pozostaw bez zmian =LICZ.JEŻELI(C:C;PRAWDA).
- Litera „a” w środku. Formuła w komórce *G4* różniła się będzie od formuł w komórkach *G2* i *G3* jedynie parametrem *zakres*, który tym razem będzie równy *D:D*. Cała formuła to =LICZ.JEŻELI(D:D;PRAWDA).

	A	B	C	D	E	F	G
1	wyraz	pierwsza	ostatnia	środek			
2	adghgd	PRAWDA	FAŁSZ	FAŁSZ		"a" pierwsze	10
3	ajakok	PRAWDA	FAŁSZ	PRAWDA		"a" końcowe	13
4	akokos	PRAWDA	FAŁSZ	FAŁSZ		"a" środkowe	53
5	amkoku	PRAWDA	FAŁSZ	FAŁSZ			
6	amodda	PRAWDA	PRAWDA	FAŁSZ			
7	aokor	PRAWDA	FAŁSZ	FAŁSZ			
8	aokormn	PRAWDA	FAŁSZ	FAŁSZ			
9	apraw	PRAWDA	FAŁSZ	PRAWDA			
10	asprz	PRAWDA	FAŁSZ	FAŁSZ			